```
> restart;
> with(linalg):
```

```
Warning, the protected names norm and trace have been redefined and
unprotected
```

```
> with(inttrans):
```

```
Warning, the name hilbert has been redefined
```

```
> with(student):
```

ELSORENDU ALLANDO EGYUTTHETOS LIN. DIFF. EGYENLET RENDSZER

y1'=y2+y3+x, y2'=y1-y3+exp(2x), y3'=y1+y2-x

y1(0)=y2(0)=y3(0)=1/2

DIREKT MEGOLDAS

Matrixos alakban Y=A.y+b

```
> Y:=matrix([[diff(y1(x),x)],[diff(y2(x),x)],[diff(y3(x),x)]]);
```

$$Y := \begin{bmatrix} \frac{\partial}{\partial x}\,\mathrm{y1}(x) \\ \frac{\partial}{\partial x}\,\mathrm{y2}(x) \\ \frac{\partial}{\partial x}\,\mathrm{y3}(x) \end{bmatrix}$$

```
> A:=matrix([[0,1,1],[1,0,-1],[1,1,0]]);
```

$$A := \begin{bmatrix} 0 & 1 & 1 \\ 1 & 0 & -1 \\ 1 & 1 & 0 \end{bmatrix}$$

```
> y:=matrix([[y1(x)],[y2(x)],[y3(x)]]);
```

$$y := \begin{bmatrix} \mathrm{y1}(x) \\ \mathrm{y2}(x) \\ \mathrm{y3}(x) \end{bmatrix}$$

```
> b:=matrix([[x],[exp(2*x)],[-x]]);
```

$$b := \begin{bmatrix} x \\ e^{(2\,x)} \\ -x \end{bmatrix}$$

```
> eh:=evalm(Y=multiply(A,y));
```

$$eh := \begin{bmatrix} \frac{\partial}{\partial x}\,\mathrm{y1}(x) \\ \frac{\partial}{\partial x}\,\mathrm{y2}(x) \\ \frac{\partial}{\partial x}\,\mathrm{y3}(x) \end{bmatrix} = \begin{bmatrix} \mathrm{y2}(x) + \mathrm{y3}(x) \\ \mathrm{y1}(x) - \mathrm{y3}(x) \\ \mathrm{y1}(x) + \mathrm{y2}(x) \end{bmatrix}$$

```
> ei:=evalm(Y=multiply(A,y)+b);
```

$$ei := \begin{bmatrix} \frac{\partial}{\partial x}\,\mathrm{y1}(x) \\ \frac{\partial}{\partial x}\,\mathrm{y2}(x) \\ \frac{\partial}{\partial x}\,\mathrm{y3}(x) \end{bmatrix} = \begin{bmatrix} \mathrm{y2}(x) + \mathrm{y3}(x) + x \\ \mathrm{y1}(x) - \mathrm{y3}(x) + e^{(2\,x)} \\ \mathrm{y1}(x) + \mathrm{y2}(x) - x \end{bmatrix}$$

Homogen megoldasa:

```
> s:=eigenvects(A);
```

$$s := [-1,\ 1,\ \{[-1,\ 1,\ 0]\}],\ [1,\ 1,\ \{[1,\ 0,\ 1]\}],\ [0,\ 1,\ \{[1,\ -1,\ 1]\}]$$

Az elso elem a sajatertek, a masodik a multiplicitasa, a harmadik a sajatvektor., tehat

```
> s1:=s[1][1];
```

$$s1 := -1$$

```
> s2:=s[2][1];
```

$$s2 := 1$$

```
> s3:=s[3][1];
```

$$s3 := 0$$

```
> v1:=s[1][3][1];
```

$$v1 := [-1,\ 1,\ 0]$$

```
> v2:=s[2][3][1];
```

$$v2 := [1,\ 0,\ 1]$$

```
> v3:=s[3][3][1];
```

$$v3 := [1,\ -1,\ 1]$$

```
> V1:=convert(v1,matrix);
```

$$V1 := \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$$

```
> V2:=convert(v2,matrix);
```

$$V2 := \begin{bmatrix} 1 \\ 0 \\ 1 \end{bmatrix}$$

```
> V3:=convert(v3,matrix);
```

$$V3 := \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix}$$

```
> e1:=evalm(exp(s1*x)*V1);
```

$$e1 := \begin{bmatrix} -e^{(-x)} \\ e^{(-x)} \\ 0 \end{bmatrix}$$

```
> e2:=evalm(exp(s2*x)*V2);
```

$$e2 := \begin{bmatrix} e^x \\ 0 \\ e^x \end{bmatrix}$$

```
> e3:=evalm(exp(s3*x)*V3);
```

$$e3 := \begin{bmatrix} 1 \\ -1 \\ 1 \end{bmatrix}$$

Tehat az alaprendszer:

```
> F:=evalm(augment(e1,e2,e3));
```

$$F := \begin{bmatrix} -e^{(-x)} & e^x & 1 \\ e^{(-x)} & 0 & -1 \\ 0 & e^x & 1 \end{bmatrix}$$

A homogen altalanos: y_ha = F.c

```
> c:=matrix([[c1],[c2],[c3]]);
```

$$c := \begin{bmatrix} c1 \\ c2 \\ c3 \end{bmatrix}$$

```
> Z:=multiply(F,c);
```

$$Z := \left[\begin{array}{c} -e^{(-x)}\, c1 + e^x\, c2 + c3 \\ e^{(-x)}\, c1 - c3 \\ e^x\, c2 + c3 \end{array}\right]$$

azaz

```
> Y1:=Z[1,1];
```

$$Y1 := -e^{(-x)}\, c1 + e^x\, c2 + c3$$

```
> Y2:=Z[2,1];
```

$$Y2 := e^{(-x)}\, c1 - c3$$

```
> Y3:=Z[3,1];
```

$$Y3 := e^x\, c2 + c3$$

Ellenorzes: behelyettesitjuk a homogen egyenletbe:

```
> evalm(subs(y1(x)=Y1,y2(x)=Y2,y3(x)=Y3,eh));
```

$$\left[\begin{array}{c} \frac{\partial}{\partial x}\left(-e^{(-x)}\, c1 + e^x\, c2 + c3\right) \\ \frac{\partial}{\partial x}\left(e^{(-x)}\, c1 - c3\right) \\ \frac{\partial}{\partial x}\left(e^x\, c2 + c3\right) \end{array}\right] = \left[\begin{array}{c} e^{(-x)}\, c1 + e^x\, c2 \\ -e^{(-x)}\, c1 \\ e^x\, c2 \end{array}\right]$$

Inhomogen, allandok varialasa: y_ia = F. d(x), ahol d(x) ugy kaphato, hogy
megoldandjuk az F.d'(x)=b egyenletrendszert.

Az egyenletrendszer matrixa:

```
> M:=evalm(augment(F,b));
```

$$M := \left[\begin{array}{cccc} -e^{(-x)} & e^x & 1 & x \\ e^{(-x)} & 0 & -1 & e^{(2\,x)} \\ 0 & e^x & 1 & -x \end{array}\right]$$

Gauss eliminacio:

```
> MM:=simplify(expand(gaussjord(M)));
```

$$MM := \left[\begin{array}{cccc} 1 & 0 & 0 & -2\,x\,e^x \\ 0 & 1 & 0 & \left(x + e^{(2\,x)}\right) e^{(-x)} \\ 0 & 0 & 1 & -2\,x - e^{(2\,x)} \end{array}\right]$$

Innen leolvashatoak a d1(x), d2(x), d3(x) derivaltjai, tehat megkaphatoak maguk a d1, d2, d3 :

```
> d1:=simplify(expand(subs(t=x,int(MM[1,4],x=0..t))));
```

$$d1 := -2\,x\,e^x + 2\,e^x - 2$$

```
> d2:=simplify(expand(subs(t=x,int(MM[2,4],x=0..t))));
```

$$d2 := -e^{(-x)}\,x - e^{(-x)} + e^x$$

```
> d3:=simplify(expand(subs(t=x,int(MM[3,4],x=0..t))));
```

$$d3 := -x^2 - \frac{1}{2}\,e^{(2\,x)} + \frac{1}{2}$$

Vektort csinalunk a d-bol, hogy az alaprendszer matrixaval osszeszorozhassuk
(ez csak Maple ugyeskedes):

```
> d:=convert([d1,d2,d3],vector);
```

$$d := \left[-2\,x\,e^x + 2\,e^x - 2,\; -e^{(-x)}\,x - e^{(-x)} + e^x,\; -x^2 - \frac{1}{2}\,e^{(2\,x)} + \frac{1}{2}\right]$$

```
> dd:=simplify(evalm(augment(d)));
```

$$dd := \begin{bmatrix} -2\,x\,e^x + 2\,e^x - 2 \\ -e^{(-x)}\,x - e^{(-x)} + e^x \\ -x^2 - \frac{1}{2}\,e^{(2\,x)} + \frac{1}{2} \end{bmatrix}$$

Es ime a partikularis megoldas vektora:

```
> S:=simplify(expand(multiply(F,dd)));
```

$$S := \begin{bmatrix} x - \frac{5}{2} + 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} - x^2 \\ -2\,x + \frac{3}{2} - 2\,e^{(-x)} + x^2 + \frac{1}{2}\,e^{(2\,x)} \\ -\frac{1}{2} - x + \frac{1}{2}\,e^{(2\,x)} - x^2 \end{bmatrix}$$

Es ezzel az inhomogen partikularis

```
> w1:=S[1,1];
```

$$w1 := x - \frac{5}{2} + 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} - x^2$$

```
> w2:=S[2,1];
```

$$w2 := -2\,x + \frac{3}{2} - 2\,e^{(-x)} + x^2 + \frac{1}{2}\,e^{(2\,x)}$$

```
> w3:=S[3,1];
```

$$w3 := -\frac{1}{2} - x + \frac{1}{2}\,e^{(2\,x)} - x^2$$

Ellenorzes: behelyettesitjuk az inhomogen egyenletbe:

```
> evalm(subs(y1(x)=w1,y2(x)=w2,y3(x)=w3,ei));
```

$$\left[\begin{array}{c} \frac{\partial}{\partial x}\,(x - \frac{5}{2} + 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} - x^2) \\ \frac{\partial}{\partial x}\,(-2\,x + \frac{3}{2} - 2\,e^{(-x)} + x^2 + \frac{1}{2}\,e^{(2\,x)}) \\ \frac{\partial}{\partial x}\,(-\frac{1}{2} - x + \frac{1}{2}\,e^{(2\,x)} - x^2) \end{array}\right] = \left[\begin{array}{c} -2\,x + 1 - 2\,e^{(-x)} + e^{(2\,x)} \\ 2\,x - 2 + 2\,e^{(-x)} + e^{(2\,x)} \\ -1 + e^{(2\,x)} - 2\,x \end{array}\right]$$

Tehat az altalanos megoldas:

```
> z1:=Y1+w1;
```

$$z1 := -e^{(-x)}\,c1 + e^x\,c2 + c3 + x - \frac{5}{2} + 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} - x^2$$

```
> z2:=Y2+w2;
```

$$z2 := e^{(-x)}\,c1 - c3 - 2\,x + \frac{3}{2} - 2\,e^{(-x)} + x^2 + \frac{1}{2}\,e^{(2\,x)}$$

```
> z3:=Y3+w3;
```

$$z3 := e^x\,c2 + c3 - \frac{1}{2} - x + \frac{1}{2}\,e^{(2\,x)} - x^2$$

Az altalanos megoldas ellenorzese:

```
> evalm(subs(y1(x)=Y1+w1,y2(x)=Y2+w2,y3(x)=Y3+w3,ei));
```

$$\left[\begin{array}{c} \frac{\partial}{\partial x}\left(-e^{(-x)}\,c1+e^{x}\,c2+c3+x-\frac{5}{2}+2\,e^{(-x)}+\frac{1}{2}\,e^{(2\,x)}-x^2\right) \\ \frac{\partial}{\partial x}\left(e^{(-x)}\,c1-c3-2\,x+\frac{3}{2}-2\,e^{(-x)}+x^2+\frac{1}{2}\,e^{(2\,x)}\right) \\ \frac{\partial}{\partial x}\left(e^{x}\,c2+c3-\frac{1}{2}-x+\frac{1}{2}\,e^{(2\,x)}-x^2\right) \end{array}\right] =$$

$$\left[\begin{array}{c} e^{(-x)}\,c1-2\,x+1-2\,e^{(-x)}+e^{(2\,x)}+e^{x}\,c2 \\ -e^{(-x)}\,c1+2\,x-2+2\,e^{(-x)}+e^{(2\,x)} \\ e^{x}\,c2-1+e^{(2\,x)}-2\,x \end{array}\right]$$

Kezdet ertek: z1(0)=z2(0)=z3(0)=1/2 :

```
> simplify(subs(x=0,z1))=1/2;
```

$$-c1+c2+c3=\frac{1}{2}$$

```
> simplify(subs(x=0,z2))=1/2;
```

$$c1-c3=\frac{1}{2}$$

```
> simplify(subs(x=0,z3))=1/2;
```

$$c2+c3=\frac{1}{2}$$

A megoldando linearis egyenletrendszer tehat:

```
> K:=matrix([[-1,1,1,1/2],[1,-1,0,1/2],[0,1,1,1/2]]);
```

$$K:=\left[\begin{array}{cccc} -1 & 1 & 1 & \frac{1}{2} \\ 1 & -1 & 0 & \frac{1}{2} \\ 0 & 1 & 1 & \frac{1}{2} \end{array}\right]$$

```
> KV:=gaussjord(K);
```

$$KV := \begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & \frac{-1}{2} \\ 0 & 0 & 1 & 1 \end{bmatrix}$$

Ezzel megvannak az alkalmas konstansok, igy a kezdeti ertek problema megoldasa :

```
> W1:=subs(c1=0,c2=-1/2,c3=1,z1);
```

$$W1 := -\frac{3}{2} - \frac{1}{2}e^x + x + 2e^{(-x)} + \frac{1}{2}e^{(2x)} - x^2$$

```
> W2:=subs(c1=0,c2=-1/2,c3=1,z2);
```

$$W2 := \frac{1}{2} - 2x - 2e^{(-x)} + x^2 + \frac{1}{2}e^{(2x)}$$

```
> W3:=subs(c1=0,c2=-1/2,c3=1,z3);
```

$$W3 := -\frac{1}{2}e^x + \frac{1}{2} - x + \frac{1}{2}e^{(2x)} - x^2$$

```
> restart;
> with(linalg):
```

```
Warning, the protected names norm and trace have been redefined and
unprotected
```

```
> with(inttrans):
```

```
Warning, the name hilbert has been redefined
```

```
> with(student):
```

MEGOLDAS LAPLACE-val

```
> de1:=diff(y1(x),x)=y2(x)+y3(x)+x;
```

$$de1 := \frac{\partial}{\partial x}\,y1(x) = y2(x) + y3(x) + x$$

```
> de2:=diff(y2(x),x)=y1(x)-y3(x)+exp(2*x);
```

$$de2 := \frac{\partial}{\partial x}\,y2(x) = y1(x) - y3(x) + e^{(2x)}$$

```
> de3:=diff(y3(x),x)=y1(x)+y2(x)-x;
```

$$de3 := \frac{\partial}{\partial x}\,y3(x) = y1(x) + y2(x) - x$$

Ezek Laplace transzformaltjai:

```
> alias(U=laplace(y1(x),x,s));
```

$$Point,\, U$$

```
> alias(V=laplace(y2(x),x,s));
```

$$Point,\, U,\, V$$

```
> alias(W=laplace(y3(x),x,s));
```

$$Point,\, U,\, V,\, \mathrm{W}$$

```
> alias(a=y1(0));
```

$$Point,\, U,\, V,\, \mathrm{W},\, a$$

```
> alias(b=y2(0));
```

$$Point,\, U,\, V,\, \mathrm{W},\, a,\, b$$

```
> alias(c=y3(0));
```

$$Point,\, U,\, V,\, \mathrm{W},\, a,\, b,\, c$$

```
> L1:=laplace(de1,x,s);
```

$$L1 := s\,U - a = V + \mathrm{W} + \frac{1}{s^2}$$

```
> L2:=laplace(de2,x,s);
```

$$L2 := s\,V - b = U - \mathrm{W} + \frac{1}{s-2}$$

```
> L3:=laplace(de3,x,s);
```

$$L3 := s\,\mathrm{W} - c = U + V - \frac{1}{s^2}$$

Most a Laplace transzformaltakbol allo linearis egyenletrendszert kell megoldani:

```
> G:=solve({L1,L2,L3},{U,V,W});
```

$$G := \{V = \frac{b\,s^4 + s^3 - c\,s^3 - b\,s^3 + a\,s^3 - 2\,b\,s^2 + 2\,c\,s^2 - 2\,a\,s^2 + s^2 + 2\,s - 4}{(s+1)\,(s-2)\,s^3},$$
$$\mathrm{W} = \frac{a\,s^3 - 3\,c\,s^3 + 4\,s - 2\,a\,s^2 + 2\,c\,s^2 - 4 + c\,s^4 + b\,s^3 - 2\,b\,s^2}{s^3\,(-3\,s + 2 + s^2)}, U = (-4 - 3\,c\,s^3$$
$$+ a\,s^3 - 2\,b\,s^2 - b\,s^3 + c\,s^4 + b\,s^4 - 2\,a\,s^4 + a\,s^5 + 4\,s + 2\,s^3 - 2\,s^2 + 2\,c\,s^2 - 2\,a\,s^2$$
$$)/(s^3\,(-3\,s + 2 + s^2)\,(s+1))\}$$

Egyenletrendszer megoldas ellenorzes:

```
> A:=matrix([[s,-1,-1,a+1/s^2],[-1,s,1,b+1/(s-2)],[-1,-1,s,c-1/s^2]]):
```

```
> gaussjord(A):
```

Parcialis tortekre bontva (ez persze itt nem kell, csak ha kezzel szamoljuk
a visszatranszformaltat):

```
> convert(G[1],parfrac,s);
```

$$V = \frac{-2 + c - a}{s+1} + \frac{\frac{1}{2}}{s-2} + \frac{\frac{1}{2}\,(2\,a + 3 + 2\,b - 2\,c)}{s} - \frac{2}{s^2} + \frac{2}{s^3}$$

```
> convert(G[2],parfrac,s);
```

$$\mathrm{W} = \frac{b+a}{-1+s} + \frac{\frac{1}{2}}{s-2} - \frac{1}{2}\,\frac{1 - 2\,c + 2\,b + 2\,a}{s} - \frac{1}{s^2} - \frac{2}{s^3}$$

```
> convert(G[3],parfrac,s);
```

$$U = -\frac{-2 + c - a}{s+1} - \frac{1}{2}\,\frac{5 - 2\,c + 2\,a + 2\,b}{s} + \frac{1}{s^2} - \frac{2}{s^3} + \frac{b+a}{-1+s} + \frac{\frac{1}{2}}{s-2}$$

Ezeket visszatranszformalva:

```
> z1:=invlaplace(G[1],s,x);
```

$$z1 := \mathrm{y2}(x) = (-2 + c - a)\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} + a + \frac{3}{2} + b - c - 2\,x + x^2$$

```
> z2:=invlaplace(G[2],s,x);
```

$$z2 := \mathrm{y3}(x) = -b - a + c - \frac{1}{2} - x - x^2 + (b + a)\, e^x + \frac{1}{2}\, e^{(2\,x)}$$

```
> z3:=invlaplace(G[3],s,x);
```

$$z3 := \mathrm{y1}(x) = -\frac{5}{2} + c - a - b + x - x^2 + (b + a)\, e^x + \frac{1}{2}\, e^{(2\,x)} + (a + 2 - c)\, e^{(-x)}$$

Ahhoz, hogy a szokasos jelolessel (amit a kozvetlen megoldasnal hasznaltunk)
kapjuk a megoldast, az egyutthatokat alkalmasan atjeloljuk:

```
> e1:=c_1=-5/2-b+c-a+3;
```

$$e1 := c\_1 = \frac{1}{2} - b + c - a$$

```
> e2:=c_2=b+a;
```

$$e2 := c\_2 = b + a$$

```
> e3:=c_3=2-c+a;
```

$$e3 := c\_3 = a + 2 - c$$

```
> s:=solve({e1,e2,e3},{a,b,c});
```

$$s := \{a = c\_2 + c\_3 - \frac{5}{2} + c\_1,\, b = -c\_3 + \frac{5}{2} - c\_1,\, c = c\_1 - \frac{1}{2} + c\_2\}$$

```
> s[1];
```

$$a = c\_2 + c\_3 - \frac{5}{2} + c\_1$$

```
> s[2];
```

$$b = -c\_3 + \frac{5}{2} - c\_1$$

```
> s[3];
```

$$c = c\_1 - \frac{1}{2} + c\_2$$

Ezeket visszahelyettesitve a megoldasba megkapjuk az altalanos megoldast (valamiert nem okkvetlenul sorremdben adja meg):

```
> m1:=subs(a=c_2+c_3-5/2+c_1,b=-c_3+5/2-c_1,c=c_1-1/2+c_2,z1);
```

$$m1 := \mathrm{y2}(x) = -c\_3\, e^{(-x)} + \frac{1}{2}\, e^{(2\,x)} + 2 - c\_1 - 2\,x + x^2$$

```
> m2:=subs(a=c_2+c_3-5/2+c_1,b=-c_3+5/2-c_1,c=c_1-1/2+c_2,z2);
```

$$m2 := \mathrm{y3}(x) = -1 + c\_1 - x - x^2 + c\_2\, e^x + \frac{1}{2}\, e^{(2\,x)}$$

```
> m3:=subs(a=c_2+c_3-5/2+c_1,b=-c_3+5/2-c_1,c=c_1-1/2+c_2,z3);
```

$$m3 := \mathrm{y1}(x) = c\_1 - 3 + x - x^2 + c\_2\, e^x + \frac{1}{2}\, e^{(2\,x)} + c\_3\, e^{(-x)}$$

(b) Kezdeti ertek problema: y1(0)=y2(0)=y3(0)=1/2

A kezdeti ertek problema megoldasat termeszetesen a z1, z2, z3-bol
az a=b=c=1/2
helyettesitesekbol kovetlenul megkaphatjuk:

```
> k1:=subs(a=1/2, b=1/2, c=1/2, z1);
```

$$k1 := \mathrm{y2}(x) = -2\, e^{(-x)} + \frac{1}{2}\, e^{(2\,x)} + 2 - 2\,x + x^2$$

```
> k2:=subs(a=1/2, b=1/2, c=1/2, z2);
```

$$k2 := \mathrm{y3}(x) = -1 - x - x^2 + e^x + \frac{1}{2}\, e^{(2\,x)}$$

```
> k3:=subs(a=1/2, b=1/2, c=1/2, z3);
```

$$k3 := \mathrm{y1}(x) = -3 + x - x^2 + e^x + \frac{1}{2}\, e^{(2\,x)} + 2\, e^{(-x)}$$

De persze kozvetlenul az elejen is elvegezhettuk
volna a helyettesitest:

```
> restart;
> with(linalg):

Warning, the protected names norm and trace have been redefined and
unprotected

> with(inttrans):

Warning, the name hilbert has been redefined

> with(student):
```

```
> de1:=diff(y1(x),x)=y2(x)+y3(x)+x;
```

$$de1 := \frac{\partial}{\partial x}\, \mathrm{y1}(x) = \mathrm{y2}(x) + \mathrm{y3}(x) + x$$

```
> de2:=diff(y2(x),x)=y1(x)-y3(x)+exp(2*x);
```

$$de2 := \frac{\partial}{\partial x}\, \mathrm{y2}(x) = \mathrm{y1}(x) - \mathrm{y3}(x) + e^{(2\,x)}$$

```
> de3:=diff(y3(x),x)=y1(x)+y2(x)-x;
```

$$de3 := \frac{\partial}{\partial x}\,\mathrm{y3}(x) = \mathrm{y1}(x) + \mathrm{y2}(x) - x$$

Ezek Laplace transzformaltjai:

```
> alias(U=laplace(y1(x),x,s));
```

$$Point,\ U$$

```
> alias(V=laplace(y2(x),x,s));
```

$$Point,\ U,\ V$$

```
> alias(W=laplace(y3(x),x,s));
```

$$Point,\ U,\ V,\ \mathrm{W}$$

```
> alias(a=y1(0));
```

$$Point,\ U,\ V,\ \mathrm{W},\ a$$

```
> alias(b=y2(0));
```

$$Point,\ U,\ V,\ \mathrm{W},\ a,\ b$$

```
> alias(c=y3(0));
```

$$Point,\ U,\ V,\ \mathrm{W},\ a,\ b,\ c$$

```
> L1:=laplace(de1,x,s);
```

$$L1 := s\,U - a = V + \mathrm{W} + \frac{1}{s^2}$$

```
> L2:=laplace(de2,x,s);
```

$$L2 := s\,V - b = U - \mathrm{W} + \frac{1}{s-2}$$

```
> L3:=laplace(de3,x,s);
```

$$L3 := s\,\mathrm{W} - c = U + V - \frac{1}{s^2}$$

```
> k1:=subs(a=1/2, b=1/2, c=1/2, L1);
```

$$k1 := s\,U - \frac{1}{2} = V + \mathrm{W} + \frac{1}{s^2}$$

```
> k2:=subs(a=1/2, b=1/2, c=1/2, L2);
```

$$k2 := s\,V - \frac{1}{2} = U - \mathrm{W} + \frac{1}{s-2}$$

```
> k3:=subs(a=1/2, b=1/2, c=1/2, L3);
```

$$k3 := s\,\mathrm{W} - \frac{1}{2} = U + V - \frac{1}{s^2}$$

Most a Laplace transzformaltakbol allo linearis egyenletrendszert kell megoldani:

```
> G:=solve({k1,k2,k3},{U,V,W});
```

$$G := \{U = \frac{1}{2}\,\frac{s^5+s^3-6\,s^2+8\,s-8}{(1+s)\,(s-2)\,(s-1)\,s^3},\ V = \frac{1}{2}\,\frac{s^4+s^3+4\,s-8}{s^3\,(s-2)\,(1+s)},$$
$$\mathrm{W} = \frac{1}{2}\,\frac{-8-s^3+8\,s-2\,s^2+s^4}{s^3\,(2-3\,s+s^2)}\}$$

Egyenletrendszer megoldas ellenorzes:

```
> A:=matrix([[s,-1,-1,a+1/s^2],[-1,s,1,b+1/(s-2)],[-1,-1,s,c-1/s^2]]):
```

```
> gaussjord(A):
```

Parcialis tortekre bontva (ez persze itt nem kell, csak ha kezzel
szamoljuk
a visszatranszformaltat):

```
> convert(G[1],parfrac,s);
```

$$U = 2\,\frac{1}{1+s} + \frac{\frac{1}{2}}{s-2} + \frac{1}{s-1} - \frac{2}{s^3} + \frac{1}{s^2} - \frac{3}{s}$$

```
> convert(G[2],parfrac,s);
```

$$V = -2\,\frac{1}{1+s} + \frac{\frac{1}{2}}{s-2} + \frac{2}{s^3} - \frac{2}{s^2} + \frac{2}{s}$$

```
> convert(G[3],parfrac,s);
```

$$\mathrm{W} = \frac{1}{2}\,\frac{1}{s-2} + \frac{1}{s-1} - \frac{2}{s^3} - \frac{1}{s^2} - \frac{1}{s}$$

Ezeket visszatranszformalva:

```
> z1:=invlaplace(G[1],s,x);
```

$$z1 := \mathrm{y1}(x) = 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} + e^x - x^2 + x - 3$$

```
> z2:=invlaplace(G[2],s,x);
```

$$z2 := \mathrm{y2}(x) = x^2 - 2\,x + 2 + \frac{1}{2}\,e^{(2\,x)} - 2\,e^{(-x)}$$

```
> z3:=invlaplace(G[3],s,x);
```

$$z3 := \mathrm{y3}(x) = -x^2 - x - 1 + e^x + \frac{1}{2}\,e^{(2\,x)}$$

ELLENORZES

Altalanos megoldas:

```
> ds:=dsolve({de1,de2,de3},{y1(x),y2(x),y3(x)});
```

$$ds := \{\mathrm{y3}(x) = e^x\,\_C2 + 2 - x - x^2 + \_C3 + \frac{1}{2}\,e^{(2\,x)},$$
$$\mathrm{y2}(x) = e^{(-x)}\,\_C1 - 2\,x + \frac{1}{2}\,e^{(2\,x)} - 1 + x^2 - \_C3,$$
$$\mathrm{y1}(x) = e^x\,\_C2 - e^{(-x)}\,\_C1 - x^2 + \frac{1}{2}\,e^{(2\,x)} + x + \_C3\}$$

Tehat:

```
> ds[1];
```

$$\mathrm{y3}(x) = e^x\,\_C2 + 2 - x - x^2 + \_C3 + \frac{1}{2}\,e^{(2\,x)}$$

```
> ds[2];
```

$$\mathrm{y2}(x) = e^{(-x)}\,\_C1 - 2\,x + \frac{1}{2}\,e^{(2\,x)} - 1 + x^2 - \_C3$$

```
> ds[3];
```

$$\mathrm{y1}(x) = e^x\,\_C2 - e^{(-x)}\,\_C1 - x^2 + \frac{1}{2}\,e^{(2\,x)} + x + \_C3$$

Az osszehasonlitashoz (lehet, hogy maskor mas sorrendben sorolja fel,
ugyhogy az indexek valtozhatnak, meg kell nezni, hogy melyik komponens
melyik megoldast adja). A megfelelo c-k megvalasztasaval lathatoan)

azonos alakuak.

Es a kezdeti ertek problema megoldasanak ellenorzese:

```
> dsk:=dsolve({de1,de2,de3,y1(0)=1/2,y2(0)=1/2,y3(0)=1/2},
> {y1(x),y2(x),y3(x)});
```

$$dsk := \{\mathrm{y1}(x) = 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} + e^x - x^2 + x - 3,\ \mathrm{y2}(x) = x^2 - 2\,x + 2 + \frac{1}{2}\,e^{(2\,x)} - 2\,e^{(-x)},$$
$$\mathrm{y3}(x) = -x^2 - x - 1 + e^x + \frac{1}{2}\,e^{(2\,x)}\}$$

```
> dsk[1];
```

$$\mathrm{y1}(x) = 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} + e^x - x^2 + x - 3$$

```
> dsk[2];
```

$$\mathrm{y2}(x) = x^2 - 2\,x + 2 + \frac{1}{2}\,e^{(2\,x)} - 2\,e^{(-x)}$$

```
> dsk[3];
```

$$\mathrm{y3}(x) = -x^2 - x - 1 + e^x + \frac{1}{2}\,e^{(2\,x)}$$

Es az osszehasonlitas (ez konnyebb):

```
> z1;
```

$$\mathrm{y1}(x) = 2\,e^{(-x)} + \frac{1}{2}\,e^{(2\,x)} + e^x - x^2 + x - 3$$

```
> z2;
```

$$\mathrm{y2}(x) = x^2 - 2\,x + 2 + \frac{1}{2}\,e^{(2\,x)} - 2\,e^{(-x)}$$

```
> z3;
```

$$\mathrm{y3}(x) = -x^2 - x - 1 + e^x + \frac{1}{2}\,e^{(2\,x)}$$